
Abstract
In this paper we present CaffeLink an open-source library for Mathematica which is a binding of a
well-established Caffe deep learning framework. Caffe is a highly-optimized CUDA accelerated
library with focus on convolutional neural networks written in C++ with Python and Matlab
bindings. CaffeLink is based upon Mathematica’s LibraryLink. It makes accessible most features of
Caffe directly from Mathematica environment which includes work with datasets, building
networks, training them as well as evaluating them. Here we present an overview of the CaffeLink
library with examples on MNIST and ImageNet datasets.

Introduction
Deep learning [A, B] describes a set of data processing algorithms based on multistage non-linear transformations. Among others it has
been successfully used for state-of-the-art applications in image processing [C], object detection [D], audio processing [E], natural
language processing [F], speech transcription [G]. The idea of deep learning is inspired in Nature where brains of living creatures process
raw data (features) by successive transformations while increasing the level of abstraction at the same time [H, I]. In deep learning such
stacked transformations are represented by deep architectures of Artificial Neural Networks (ANNs) or related paradigms, e.g., Restricted
Boltzmann Machines (RBMs) [J]. These architectures are often called Deep Neural Networks (DNNs). In most cases DNNs are learned
using supervised gradient descent approaches, although other methods such as Evolutionary Algorithms were also employed [K].
In most practical applications DNNs are very large: for ImageNet image recognition dataset, which we present as an example in this paper,
a Convolutional Neural Network (CNN) model contains about 500,000 neurons and approximately 60 millions parameters have to be
optimized in order to get a classifier based on 1.2 million images. One can see that both learning and evaluation (application of all
transforms on an input pattern to get an output) of such architectures is computationally highly demanding. Nevertheless most algorithms
can be efficiently run in parallel on modern GPUs.
Currently there exist several implementations of DNNs with GPU support [L, M, N, O]. Unfortunately none of them has a Mathematica
interface. Although Mathematica has a good support for GPU computing (both CUDA and OpenCL) we have decided not to write our
solution from the scratch but rather base it on an existing solution called Caffe [L, P]. Caffe is a highly-optimized implementation of
DNNs (authors report processing speeds of more than 40 million images per day on a single K40 or Titan GPU) which supports both CPU
and GPU computations. It covers the whole work-flow: building model architectures, training, model evaluation and deployment. Caffe is
written in C++ and it has currently bindings for Python and Matlab environments.
We call the resulting library CaffeLink. It is a LibraryLink based Mathematica binding to Caffe licensed using BSD 2-Clause license.
CaffeLink can be downloaded at https://github.com/Seilim/CaffeLink where you can find more documentation as well as more detailed
Mathematica example notebooks.
This paper does cover neither deep learning nor Caffe in detail, we refer the reader to [A] and [L, P] for more information. It is organized
as follows: in the next section we present CaffeLink Architecture. Then a section describing the most important CaffeLink library
functions follows. The last two sections show CaffeLink in action on MNIST and ImageNet datasets.

CaffeLink Architecture
CaffeLink architecture is depicted in Figure 1. The communication between Caffe and CaffeLink is realized by Wolfram LibraryLink. All
computations on DNNs (DNN, network or model are used interchangeably throughout the following text) including training and network
evaluation can be performed either on CPU or GPU.

Figure 1: Caffe Link Architecture. Red color indicates Mathematica-based components including CaffeLink, orange color Caffe performing
computations either on CPU or GPU. Blue components represent input/output data. Disk icons represent data stored/retrieved from filesystem.

In order to learn or evaluate a DNN one must supply Caffe with the model prototype describing an architecture of the network (see below
for more details). There are two possibilities of how to define the model prototype. First it can be generated directly from a Mathematica
expression using CaffeLink. Second a Caffe configuration file can be imported using Google Protocol Buffer format (protobuf).
For training, Caffe must be supplied a solver configuration. Similarly to model prototypes, both Mathematica expression and configuration
file are supported. Caffe currently supports three solver methods: Stochastic Gradient Descent (SGD), Adaptive Gradient (ADAGRAD)
and Nesterov’s Accelerated Gradient (NAG). See https://github.com/BVLC/caffe/blob/master/docs/tutorial/solver.md for more details.
For efficiency, training datasets can be supplied only using standard paths used by Caffe which are: LevelDB and LMDB databases, direct
memory access or HDF5 files. Model evaluation can be additionally realized directly from Mathematica by means of real-valued vectors.
Based on the solver configuration, during and after training, Caffe can store a model snapshot which is basically a set of all model’s
parameters (i.e., synaptic connection weights and neuron thresholds). Having both model prototype and model snapshot, the model can be
deployed in memory and evaluated. Similarly a solver state can be stored in a filesystem which allows to resume the learning process later.
Model prototypes are defined in a layer-by-layer fashion. Definition of each layer involves type, name, inputs, outputs and other parameters. Caffe supports a large set of layer types see https://github.com/BVLC/caffe/blob/master/docs/tutorial/layers.md for more details. The
following excerpt written in Google Protocol Buffer format shows an example of a convolutional layer followed by a rectified linear unit
(ReLU) layer:
layers {
name : "conv1"
type : CONVOLUTION
bottom : "data"
top : "conv1"
convolution_param {
num_output: 96
kernel_size: 11
stride : 4
}
}
layers {
name : "relu1"
type : RELU
bottom : "conv1"
top : "relu1"
}

The input of the “conv1” convolutional layer is designated “data” which is a name of a preceding layer blob (not shown). The “conv1”
layer has 96 filters of size 11x11. ReLU “relu1” layer follows the “conv1” layer.
Bottom and top parameters indicate input and output data blobs. Blob is Caffe’s main data structure used to move data between layers and
to set their parameters. Each blob is represented by a one-dimensional array. Its logical structure is, however, four-dimensional: it is
defined by four coordinates (N x C x H x W). If the blob represents an image dataset (which is often the case) then the quadruple can be
interpreted as (the number of images x the number of color channels x image height x image width). Blob can represent any type of data.
Its internal representation is important only in a case of direct data input or output from or to Caffe (which is realized by many CaffeLink
functions).

CaffeLink Functions
In this section we describe the most important CaffeLink functions. Note that it is a good idea to start Mathematica from console/command
line as Caffe outputs a lot of useful information to the console. The functions are summarized in the following Tables 1 through 7:
Table 1: Initialization.
Name

Parameters

Notes

initCaffeLinkModule

path to caffe.proto file from Caffe installation

Initializes CaffeLink module.

initCaffeLink

UseDouble, UseGPU, GPUDevice

Initializes CaffeLink library, sets computation mode and floating data type precision.

Table 2: Initializing model prototypes and solver configurations.
Name

Parameters

Notes

newNet

model prototype as a Mathematica expression

Returns CaffeLink internal parameter representation.

newSolver

solver configuration as a Mathematica expression

Returns CaffeLink internal parameter representation.

getParamString

parameters

Returns a protobuf string representation of parameters
created with newNet or newSolver.

solverAddNetParam

solver parameters, model parameters

Combines solver configuration and model prototype (both in protobuf format).

prepareNetFile

path

Sets up a new network from protobuf file on given path.

prepareNetString

string

Sets up a new network from given protobuf string.

Table 3: Model evaluation.
Name

Parameters

Notes

setInput

data

Loads the input blob with ' data '. Works with networks without input data layers only.

evaluateNet

Evaluates /∕ activates the network.

Table 4: Model training.
Name

Parameters

Notes

trainNetString

string

Trains network with solver protobuf formatted parameters given as a string.

trainNetFile

path

Trains network with solver protobuf formatted parameters taken from file.

trainNetSnapshotString

string, pathState

Continues training from solver state file on ' pathState '. Solver parameters are given as string.
Continues training from solver state file on ' pathState '. Solver parameters given as a ' path ' to a file.

trainNetSnapshotFile

path, pathState

trainNetWeightsString

string, pathWeights

Finetunes network on path ' pathWeights '. Solver parameters given as a string.

trainNetWeightsFile

path, pathWeights

Finetunes network on path ' pathWeights '. Solver parameters given as a ' path ' to a file.

Table 5: Import/export functions.
Name

Parameters

Notes

loadNet

path

Loads a model snapshot after the model prototype was prepared using prepareNet functions.

exportNet

path

Exports previously loaded network to a selected path.

Table 6: Access to blobs and layers.
getLayerNum

getLayerNum

Returns the number of layers in a prepared network.

getTopBlobSize

layerIndex or name

Returns an array of the top blob dimensions : {num, channels, height, width},
4 elements for each blob.

getBottomBlobSize

layerIndex or name

The same for the bottom blob.

getParamBlobSize

layerIndex or name

The same for the parameter blob.
Returns data stored in the top blob.

getTopBlob

layerIndex or name, blobIndex

getBottomBlob

layerIndex or name, blobIndex

The same for the bottom blob.

getParamBlob

layerIndex or name, blobIndex

The same for the parameter blob.

setTopBlob

data, layerIndex or name, blobIndex

Sets the top blob.

setBottomBlob

data, layerIndex or name, blobIndex

The same for the bottom blob.

setParamBlob

data, layerIndex or name, blobIndex

The same for the parameter blob.

Table 7: Debugging functions.
Name

Parameters

Notes

printNetInfo

Prints (in a console, stdout) model prototype info : structure, layers and blob count /∕ size, etc.

printWorkingPath

Prints (in a console, stdout) a working directory as seen by Caffe. Executes ' pwd ' command.

See project web page and CaffeLink module for more details.

MNIST Dataset Example
MNIST [C] is a well-known dataset of handwritten digit images. The task is to classify the grayscale images to one of 10 classes (digits
zero to nine). The dataset contains 60,000 training and 10,000 validation examples. Image resolution is 28x28 pixels. In this section we
demonstrate how CaffeLink can be used to build a Convolutional Neural Network (CNN) model on the MNIST data. The example is
based on a similar Caffe tutorial (see http://caffe.berkeleyvision.org/gathered/examples/mnist.html for detailed explanation of layer and
solver parameters).
The CNN architecture (model prototype) is shown in Figure 2. The input grayscale image having a resolution of 28x28 is processed by the
first convolutional layer with 5x5 kernels obtaining twenty 24x24 filters. A reduction of dimensionality is realized by a max-pooling layer:
using 2x2 kernels the “pool1” layer transforms the output of the convolutional layer “conv1” to 12x12 resolution. Similar transformation is
realized by the following two layers “conv2” and “pool2” getting 50 filters of only 4x4 pixels. The following two layers “ip1” and “relu1”
represent a standard hidden layer of 500 rectified linear unit (ReLU) neurons. Similarly the last two layers “ip2” and “prob” form a
complete Softmax output layer of ten neurons indicating a probability for each of the ten classes zero to nine.

Figure 2: MNIST Convolutional Neural Network Architecture. The architecture is composed of convolutional (“conv1”, “conv2”), max-pooling
(“pool1”, “pool2”), ReLU (“ip1” and “relu1”) and Softmax (“ip2”, “prob”) layers. Note that in Caffe, ReLU and Softmax layers are both decomposed to
inner product layer and the layer computing the actual activation function. The notation 20@24x24 for “conv1” layer indicates 20 convolutional filters
of size 24x24.

Initialization
The initialization steps needed for CaffeLink are shown below. Note that one must supply a directory containing Caffe installation. The
rule “UseGPU”→False indicates that CPU will be used for the computations (switch to True for CUDA capable hardware).
caffeDir = "...";
Needs["CaffeLink`"]
initCaffeLinkModule[FileNameJoin[{caffeDir,"src/∕caffe/∕proto/∕caffe.proto"}]];
initCaffeLink["UseDouble"→ True,"UseGPU"→False,"GPUDevice"→ 0];
printWorkingPath[];

Model Prototype Creation
Training Model Prototype
The code below generates a string representation of the network as depicted in Figure 2. There are two data type input layers for both
training and testing phases. There is also an additional “accuracy” layer for the testing phase. Here we show a Mathematica expression
version of the model prototype. The newNet function converts the expression to CaffeLink internal representation while getParamString
function gives the representation in the protobuf format. Resulting string is equal to the content of lenet_train_test.prototxt which can be
found in Caffe installation directory.

netTrain =
{"name"→"Net Train",{
{
"name"→"mnist","type"→"DATA","top"→{"data","label"},
"source"→FileNameJoin[{caffeDir,"examples/∕mnist/∕mnist_train_lmdb"}],
"backend"→"LMDB","batch_size"→64,"scale"→0.00390625,"include.phase"→"TRAIN"
},{
"name"→"mnist","type"→"DATA","top"→{"data","label"},
"source"→FileNameJoin[{caffeDir,"examples/∕mnist/∕mnist_test_lmdb"}],
"backend"→"LMDB","batch_size"→64,"scale"→0.00390625,"include.phase"→"TEST"
},{
"name"→"conv1","type"→"CONVOLUTION","bottom"→"data","top"→"conv1",
"blobs_lr"→{1,2},"num_output"→20,"kernel_size"→5,"stride"→1,
"weight_filler.type"→"xavier","bias_filler.type"→"constant"
},{
"name"→"pool1","type"→"POOLING","bottom"→"conv1","top"→"pool1",
"pool"→"MAX","kernel_size"→2,"stride"→2
},{
"name"→"conv2","type"→"CONVOLUTION","bottom"→"pool1","top"→"conv2",
"blobs_lr"→{1,2},"num_output"→50,"kernel_size"→5,"stride"→1,
"weight_filler.type"→"xavier","bias_filler.type"→"constant"
},{
"name"→"pool2","type"→"POOLING","bottom"→"conv2","top"→"pool2",
"pool"→"MAX","kernel_size"→2,"stride"→2
},{
"name"→"ip1","type"→"INNER_PRODUCT","bottom"→"pool2",
"top"→"ip1","blobs_lr"→{1,2},"num_output"→500,
"weight_filler.type"→"xavier","bias_filler.type"→"constant"
},{
"name"→"relu1","type"→"RELU","bottom"→"ip1","top"→"ip1"
},{
"name"→"ip2","type"→"INNER_PRODUCT","bottom"→"ip1",
"top"→"ip2","blobs_lr"→{1,2},"num_output"→10,
"weight_filler.type"→"xavier","bias_filler.type"→"constant"
},{
"name"→"accuracy","type"→"ACCURACY","bottom"→{"ip2","label"},
"top"→"accuracy","include.phase"→"TEST"
},{
"name"→"loss","type"→"SOFTMAX_LOSS","bottom"→{"ip2","label"},"top"→"loss"
}
}
};
netTrainParam = getParamString[newNet[netTrain]];

Solver Configuration
The following code shows how solver configuration is created. We subsequently combine both prototype and solver configurations using
solverAddNetParam. We also create “snaps” directory where the snapshots will be stored during training.
solver = newSolver[{
"solver_type"→"SGD","test_iter"→100,"test_interval"→500,
"base_lr"→0.01,"momentum"→0.9,"weight_decay"→0.0005,
"lr_policy"→"inv","gamma"→0.0001,"power"→0.75,
"display"→100,"max_iter"→1000,"snapshot"→1000,
"snapshot_prefix"→StringJoin[NotebookDirectory[],"snaps/∕lenet"],
"solver_mode"→"CPU"
}];
solverParam = getParamString[solver];
allParam = solverAddNetParam[solverParam,netTrainParam];
CreateDirectory[StringJoin[NotebookDirectory[],"snaps/∕"]];

Model Training
The training can be run by the following command. The progress can be monitored in the console.
trainNetString[allParam];

Deployment Model Prototype
CaffeLink currently does not support direct use of trained models. In order to do so one has to deploy the network. The following prepares
the model prototype for the deployment. It is the same as the training model prototype above with two exceptions. First, input is not
realized by loading dataset from database (direct access). Second, the output layers were simplified (we plan to evaluate the network only,
not to compute the loss function).

CaffeLink currently does not support direct use of trained models. In order to do so one has to deploy the network. The following prepares
the model prototype for the deployment. It is the same as the training model prototype above with two exceptions. First, input is not
realized by loading dataset from database (direct access). Second, the output layers were simplified (we plan to evaluate the network only,
not to compute the loss function).
netDeploy =
{"name"→"Net Deploy","input"→"data","input_dim"→{10,1,28,28},{
{
"name"→"conv1","type"→"CONVOLUTION","bottom"→"data","top"→"conv1",
"blobs_lr"→{1,2},"num_output"→20,"kernel_size"→5,"stride"→1,
"weight_filler.type"→"xavier","bias_filler.type"→"constant"
},{
"name"→"pool1","type"→"POOLING","bottom"→"conv1","top"→"pool1",
"pool"→"MAX","kernel_size"→2,"stride"→2
},{
"name"→"conv2","type"→"CONVOLUTION","bottom"→"pool1","top"→"conv2",
"blobs_lr"→{1,2},"num_output"→50,"kernel_size"→5,"stride"→1,
"weight_filler.type"→"xavier","bias_filler.type"→"constant"
},{
"name"→"pool2","type"→"POOLING","bottom"→"conv2","top"→"pool2",
"pool"→"MAX","kernel_size"→2,"stride"→2
},{
"name"→"ip1","type"→"INNER_PRODUCT","bottom"→"pool2","top"→"ip1",
"blobs_lr"→{1,2},"num_output"→500,
"weight_filler.type"→"xavier","bias_filler.type"→"constant"
},{
"name"→"relu1","type"→"RELU","bottom"→"ip1","top"→"ip1"
},{
"name"→"ip2","type"→"INNER_PRODUCT","bottom"→"ip1","top"→"ip2",
"blobs_lr"→{1,2},"num_output"→10,
"weight_filler.type"→"xavier","bias_filler.type"→"constant"
},{
"name"→"prob","type"→"SOFTMAX","bottom"→"ip2","top"→"prob"
}
}
};
netDeployParam = getParamString[newNet[netDeploy]];

The following two functions allocate the network and load model parameters from the snapshot:
prepareNetString[netDeployParam];
loadNet[StringJoin[NotebookDirectory[],"snaps/∕lenet_iter_1000.caffemodel"]]

The network is now ready to be evaluated on inputs. Note that the input blob dimensions were set to (10 x 1 x 28 x 28) which means the it
is configured to process 10 grayscale images of 28x28 pixels at once. The following images are taken from the MNIST testing dataset. The
network expects a flattened list of 10 x 1 x 28 x 28 = 7840 elements in the range 0 to 1 which we set in the input variable:
inputImgs = 

,

,

,

,

,

,

,

,

,

;

input = Flatten[Map[ImageData[#]&,inputImgs,1]];

Now we load the input data, activate the network and read its outputs. We use the output layer name “prob” to get the right blob.
setInput[input];
evaluateNet[];
probabilities = ArrayReshape[getTopBlob["prob"],{10,10}];

The following code generates Figure 3 which visualizes probabilities assigned by the network to each of the ten input symbols.

showProbabilities[probs_]:=
Module[{},
BarChart[probs*⋆100.0,
ChartLabels→Placed[Range[0,9],Top],
PlotLabel→"probability %",
BaseStyle→{FontSize→10},
ImageSize→150
]
]

Grid[Partition[MapThread[Grid[{{#1,showProbabilities[#2]}}]&,
{inputImgs,probabilities}],2],Frame→All]

Figure 3: MNIST model deployment. Model’s response on ten input symbols. Note that for the last six symbols there are at least two reasonably
probable classifications.

Inspecting Hidden Layers
CaffeLink allows us to inspect blobs of any layer (or any parameter set). The following code generates Figure 4 displaying activities of all
filters in layers “conv1” through “pool2” while presenting symbol “7” to the network.

layerOutput[{imageId_,layerId_,blobSize_,splitFilters_}]:=
Module[{out,outImgs,label},
out = getTopBlob[layerId];
outImgs = Map[Rescale,ArrayReshape[out,blobSize][[imageId]],1];
label = StringForm[layerId <> " ``@``x``",Sequence@@blobSize[[2;;]]];
Grid[{{label,SpanFromLeft}}~∼Join~∼
Partition[Map[Image[#,Magnification→2]&,outImgs,1],splitFilters]]
]
layers = {
{1,"conv1",{10,20,24,24},10},
{1,"pool1",{10,20,12,12},10},
{1,"conv2",{10,50,8,8},25},
{1,"pool2",{10,50,4,4},25}
};
Grid[{layerOutput[#]}& /∕@ layers,Frame→All]

conv1 20@24x24

pool1 20@12x12

conv2 50@8x8

pool2 50@4x4

Figure 4: MNIST model deployment. Hidden layer’s response on a symbol “7”.

ImageNet Dataset Example
In this section we present CaffeLink working with ImageNet classification dataset [Q] based on Large Scale Visual Recognition Challenge
2012 (ILSVRC2012). The task was to build a classifier using 1.2 million of high-resolution color images. The number of classes is 1000.
Unlike in previous MNIST example we only deploy existing reference model which is a CNN build of 23 layers (see http://caffe.berkeleyvision.org/gathered/examples/imagenet.html for more details). We load the model prototype, allocate it and load a model snapshot by the
following commands:
netDeploy = ReadString[FileNameJoin[{caffeDir, "models/∕bvlc_reference_caffenet/∕deploy.prototxt"}]];
prepareNetString[netDeploy]
loadNet[FileNameJoin[
{caffeDir,"models/∕bvlc_reference_caffenet/∕bvlc_reference_caffenet.caffemodel"}]]

The deployed prototype expects ten images as in the MNIST example. The input blob has a resolution (10 x 3 x 227 x 227) which
corresponds to ten images of three channels (Blue, Green, Red, hence the call to the Reverse function) of 227x227 pixels.

inputImgs = 

,

,

,

,

,

,

,

,

,

;

input = Flatten[ImageData /∕@
Flatten[Reverse[ColorSeparate[ImageResize[#,{227,227}]]]& /∕@ inputImgs]] *⋆ 255.0;

Loading inputs, activating the network and reading outputs is straightforward:
setInput[input];
evaluateNet[]
probabilities = ArrayReshape[getTopBlob["prob"],{10,1000}];

In order to visualize image classification we have to import class names and preprocess them:
classes = Module[{raw,noId},
raw = Import[FileNameJoin[{caffeDir, "data/∕ilsvrc12/∕synset_words.txt"}], {"Text", "Lines"}];
noId = StringSplit[#][[2 ;;]] & /∕@ raw;
StringTrim[StringJoin[#[[;;Min[Length[#],2]]]],","]& /∕@ noId
];

The following code generates Figure 5 which displays five most probable classes for each image:
classifyBestFive[probs_]:=
Module[{pairs, best},
pairs = Transpose[{probs, classes}];
best = Transpose[Reverse[Reverse[SortBy[pairs, First]][[;;5]]]];
BarChart[best[[1]]*⋆100.0,
BarOrigin→Left,
ChartLabels→Placed[best[[2]],After],
Frame→{True,False,False,False},
FrameTicks→{{None,None},{{0,25,50,75,100},None}},
PlotRange -−> {{0, 100},Automatic},
PlotRangePadding→{{0,65},0},
FrameLabel→"probability %",
BaseStyle→{FontSize→10},
ImageSize→130
]
]
Grid[Partition[MapThread[Grid[{{ImageResize[#1,130],classifyBestFive[#2]}}]&,
{inputImgs,probabilities}],2],Frame→All]

12th International Mathematica Symposium, 2015, Prague, Czech Republic

10 / 11

Figure 5: ImageNet model deployment. Network response to ten input images. Five most probable classes are shown including the probabilities.
The total number of classes is one thousand. Reference model from Caffe installation was used.

Conclusion and Future Work
In this paper we have presented CaffeLink a Mathematica binding for Caffe deep learning framework. CaffeLink is based on Mathematica’s LibraryLink. It allows to build, train and deploy Deep Neural Networks supporting acceleration on GPU. We have presented an
overview of the CaffeLink architecture, its functions and demonstrated its usage on two well-known image recognition datasets MNIST
and ImageNet.
In future we plan to add network visualization support. We also plan to improve API and extend it by higher-level constructs such as
builders for commonly used classes of DNNs.

